
Abstract

We study the properties of the uncountable set of Stewart words. These are Toeplitz
words specified by infinite sequences of Toeplitz patterns of the form αβγ, where α, β, γ
is any permutation of the symbols 0,1,?. We determine the critical exponent of the
Stewart words, prove that they avoid the pattern xxyyxx, find all factors that are
palindromes, and determine their subword complexity. An interesting aspect of our
work is that we use automata-theoretic methods and a decision procedure for automata
to carry out the proofs.

2020 AMS MSC Classifications: 11B85, 68R15, 03D05.

1 Introduction

Toeplitz words are infinite words formed by an iterative process, specified by a sequence
(ti)i≥1 of one or more Toeplitz patterns [10, 5]. A Toeplitz pattern is a finite word over the
alphabet Σ ∪ {?}, where Σ is a finite alphabet and ? is a distinguished symbol.

∗Research funded by a grant from NSERC, 2018-04118.

We start by constructing the infinite word tω1 = t1t1t1 · · · by repeating t1 infinitely. Next,
we consider the locations of the ? symbols in tω1 and replace all of the corresponding ?

symbols by tω2 , and so forth.
As an example, consider using only the single Toeplitz pattern 0?1?. At the first stage

we get
0?1?0?1?0?1?0?1?0?1?0?1?0?1?0?1? · · · .

At the second stage we get

001?011?001?011?001?011?001?011? · · · .

At the third stage we get

0010011?0011011?0010011?0011011? · · · .

At the fourth stage we get

001001100011011?001001110011011? · · · .

The limit of this process yields the infinite word

00100110001101100010011100110110 · · · ,

known as the regular paperfolding word.
On the other hand, instead of a single pattern, at each stage we could choose between

the two patterns 0?1? and 1?0?. In this case we get an uncountable set of infinite words,
called the paperfolding words. They share many of the same properties; see, for example, [6].

In this paper we consider another uncountable set of Toeplitz words that we call Stewart
words. Here we can choose, at each stage, any of the six pattern of the form αβγ, where
{α, β, γ} is some permutation of the three symbols 0,1,? and that we call Stewart patterns :

a = 01?; b = 10?;

c = 0?1; d = 1?0;

e = ?01; f = ?10.

Some choices of these patterns correspond to well-known sequences. We give two exam-
ples now.

• The pattern sequence cω = ccc · · · specifies the so-called Stewart choral sequence [17]

001001011 · · · ,

the fixed point of the morphism 0 → 001, 1 → 011. This is sequence A116178 in
the On-Line Encyclopedia of Integer Sequences (OEIS) [16]. This explains our choice
of the term “Stewart word”. It was also mentioned by Cassaigne and Karhumäki [5,
Example 4] and Berstel and Karhumäki [2, pp. 196–197].

• The pattern sequence (ab)ω = ababab · · · specifies the so-called Sierpiński gasket word

011010010 · · · ,

the fixed point of the morphism 0→ 011, 1→ 010. This is sequence A156595 in the
On-Line Encyclopedia of Integer Sequences (OEIS) [16].

• The eight “generalized choral sequences” of Noche [12, 13, 14] are (up to renaming the
first letter) the six sequences specified by aω, bω, cω, dω, eω, fω.1 For these specific
sequences, Noche proved some of the same properties we discuss here.

The Stewart words, specified by arbitrary infinite sequences of Stewart patterns in

{a, b, c, d, e, f}ω,

share many properties, as conjectured by the first author [7]. In this article we prove these
properties, and others.

An interesting feature of our proofs is that they are based on automata and expressing the
desired property in first-order logic, and then using a decision procedure to prove the result
“automatically”. We emphasize that this method permits us to handle sequences generated
by all infinite sequences of the Stewart patterns, not just the periodic ones. The same idea
was used previously to analyze the paperfolding sequences [8].

2 Notation

We define Σk = {0, 1, . . . , k− 1}. We use the concept of deterministic finite automaton with
output, the DFAO, from [1]. This is a 6-tuple (Q,Σ,∆, δ, q0, τ), where the output on input
x is τ(δ(q0, x)). For an infinite word t = t0t1t2 · · · we define t[i] = ti, and the analogous
notation for finite words. We write t[i..j] = titi+1 · · · tj.

For two infinite words t = t0t1t2 · · · and u = u0u1u2 · · · we define t×u to be the infinite
word [t0, u0][t1, u1][t2, u2] · · · , and the same for two finite words of the same length. We use
italic letters for finite words and bold-face for infinite words.

3 Finite Toeplitz words and the Stewart automaton

We start by considering finite sequences of Stewart patterns and the finite prefixes of the
Stewart words they specify.

Let t be a finite word over the alphabet A := {a, b, c, d, e, f}. Such a finite word defines
an infinite Stewart word s(t) over the alphabet B := {0,1,?} as explained above. However,

1In the case that a Stewart word is specified by an infinite word with a suffix in {e, f}ω, and only in this
case, the limit of the Toeplitz construction is an infinite word containing a single occurrence of ?. In this
special case, there are two distinct Stewart words, obtained by replacing this single occurrence of ? with 0

and 1, respectively.

we will find it useful to instead define the map T (t) to be the finite prefix of length 3|t| of s(t).
Thus, for example, T (af) = 01?011010 and T (afe) = 01?011010010011010011011010.

Alternatively, for a finite word t ∈ A∗ we can define T (t) as follows: We have T (ε) = ?;
given y = T (t), we define T (tg) for a single letter g ∈ A by replacing the three ? symbols in
y3, in order, by the three symbols T (g).

Theorem 1. There is a 3-state DFAO M over the alphabet A × Σ3 that, takes as input a
word where the first component spells out the finite sequence t of Stewart patterns, and the
second component spells out n, expressed in base 3 (with the least significant digit first), and
computes T (t)[n], the n’th symbol of T (t), where we index starting at position 0. Here the
name of the state is also its output.

Proof. The DFAO M is depicted in Figure 1. Asterisks match all possible inputs.

?

[a,2], [b,2],
[c,1], [d,1],
[e,0], [f,0] 0

[a,0], [b,1],
[c,0], [d,2],
[e,1], [f,2]

1

[a,1], [b,0],
[c,2], [d,0],
[e,2], [f,1]

[*,*]

[*,*]

Figure 1: The Stewart automaton M .

To see that the Stewart automaton works as claimed, an easy induction shows that if
n =

∑
0≤i<r ei3

i with ei ∈ Σ3, and δ is the transition function of M , then

δ(?, [t0, e0][t1, e1] · · · [tr−1, er−1]) = T (t0t1 · · · tr−1)[n].

Let us agree to call 0 and 1 boolean symbols, in contrast with the ? symbol. A factor of
a finite Toeplitz word is called boolean if all of its symbols are boolean.

4 Decision procedures and Walnut

The strategy in this paper is to express our conjectures as first-order logic formulas about
finite Toeplitz words T (t). Because we have an automaton computing T (t), these formulas
can be proved or disproved using a well-known extension of Presburger arithmetic [4].

We use Walnut [11], a free software theorem-prover for first-order statements about
automata and the sequences they compute. For more about Walnut, see https://cs.

uwaterloo.ca/~shallit/walnut.html.
Here is a brief guide to the syntax of Walnut. All variables refer to natural numbers.

• ?lsd x instructs Walnut to evaluate the formula where the default base of representa-
tion of integers is x. In our paper x can be either 3 or 7. This instruction has local
scope.

• A represents the universal quantifier ∀;

• E represents the existential quantifier ∃;

• & represents logical “and” (∧);

• | represents logical “or” (∨);

• eval evaluates a logical formula with no free variables and decides if it is true or false;

• def defines an automaton for a logical formula for later use;

• reg allows one to define an automaton from a regular expression, for later use;

• TP[t][n] represents T (t)[n].

5 Implementing the Stewart automaton in Walnut

When we implement the DFAO M in Walnut, some minor modifications are needed. Because
the outputs of automata in Walnut must be integers and not letters, we have to change the
coding of letters from a,b,c,d,e,f to numbers; we use 1, 2, 3, 4, 5, 6 instead. We do not use 0
as one of the six letter codes, in order that finite Stewart pattern sequences be allowed to end
in arbitrarily many 0’s; this permits an automaton to compare pattern sequences of different
lengths (by padding the shorter with trailing zeros) and to determine if one is a prefix of
another. Therefore, a finite Stewart pattern sequence t can be written as any element of
t0∗. The automaton M in Walnut is stored under the name TP.txt in the Word Automata

Library. We identify a finite Stewart pattern sequence with the integer it represents in
base 7, remembering that all integers in this paper are represented starting with the least
significant digit.

We also have to change the output alphabet of the automaton, replacing ? with 2. This
gives the automaton in Figure 2. A state is labeled in the form “state name/output”. Here
F = {1, 2, 3, 4, 5, 6} and G = Σ3.

0/2

[1,2], [2,2],
[3,1], [4,1],
[5,0], [6,0]

1/0

[1,0], [2,1],
[3,0], [4,2],
[5,1], [6,2]

2/1

[1,1], [2,0],
[3,2], [4,0],
[5,2], [6,1]

3/2

[0,0]

F × G

4/0[0,0]

F × G

5/1[0,0]

[0,0]

[0,0]

[0,0]

Figure 2: The automaton TP.txt in Walnut.

One of our main ideas is that to prove many statements about sequences generated by
infinite sequences of Stewart patterns, it suffices to study finite sequences of Stewart patterns
only. This is implied by the following lemma.

Lemma 2. Let t1t2 · · · be an infinite sequence of Stewart patterns. Then, for all r ≥ 1,
there exists a letter h ∈ A such that the prefix of T (t1t2 · · · tr−1h) of length 3r−1 is a prefix
of T (t1t2 · · · ).

Proof. There is only one ? symbol in T (t1t2 · · · tr−1) that can be substituted by later patterns,
and whatever substitution is made is accounted for by a letter of A.

The lemma tells us that if we study the finite sequences arising from Stewart patterns of
length r, then the first 3r−1 symbols of such sequences cover all possibilities for the prefixes
of length 3r−1 of the sequences arising from all infinite sequences of Stewart patterns.

Even more than this, we can prove the following:

Theorem 3. Let n ≥ 1, ` ≥ 2 be integers with n ≤ 3`−2, and let t ∈ A` be a length-` sequence
of Stewart codes. If t is a prefix of u, then every length-n boolean factor that appears in T (u)
also appears in T (t).

Proof. We can prove this with Walnut. We need to define various formulas useful in express-
ing the assertion of Theorem 3 in first-order logic:

• faceq(i, j, n, t) expresses the assertion that T (t)[i..i + n − 1] = T (t)[j..j + n − 1], for
integers i, j, n and a finite sequence of Stewart patterns t.

• cmp(i, j, n, t, u) expresses the assertion that T (t)[i..i+ n− 1] = T (u)[j..j + n− 1], for
integers i, j, n and finite sequences of Stewart patterns t and u.

• boolean(i, n, t) expresses the assertion that T (t)[i..i+n−1] is boolean (contains no ?).

• pref(t, u) expresses the assertion that t is a prefix of u. For the purpose of this rela-
tionship, trailing zeros in t and u are ignored.

• link(x, t) expresses the assertion that x is an integer satisfying x = 3|t|. This links the
base-3 representation of x with the base-7 representation of t.

• bnd(x, y) expresses the assertion that y = 3dlog3 xe for x ≥ 1.

• power3(x) expresses the assertion that x ≥ 1 is a power of 3.

Here is the implementation of these subroutines in Walnut.

def faceq "?lsd_3 Ak (k<n) => TP[t][i+k]=TP[t][j+k]":

# vars i,j,n,t

# 88 states, 28965 ms

def cmp "?lsd_3 Ak (k<n) => TP[t][i+k]=TP[u][j+k]":

# vars i,j,n,t,u

# 431 states, 76 Gigabytes of RAM and 2014067ms CPU

def boolean "?lsd_3 Ak (k<n) => (TP[t][i+k]=@0|TP[t][i+k]=@1)":

# vars i,n,t

# 24 states

reg pref lsd_7 lsd_7 "([1,1]|[2,2]|[3,3]|[4,4]|[5,5]|[6,6])*

([0,1]|[0,2]|[0,3]|[0,4]|[0,5]|[0,6])*[0,0]*":

# vars t1,t2

# 3 states

reg link lsd_3 lsd_7 "([0,1]|[0,2]|[0,3]|[0,4]|[0,5]|[0,6])*[1,0][0,0]*":

# vars x,t

# 2 states

reg bnd lsd_3 lsd_3 "([0,0]|[1,0]|[2,0])*[0,1][0,0]*|[0,0]*[1,1][0,0]*":

# vars x,y

# 3 states

reg power3 lsd_3 "0*10*":

# var x

Finally, we translate the statement of Theorem 3 into Walnut. Instead of representing |t|
(resp., |u|), we use the variable x (resp., y) to represent 3|t| (resp., 3|u|).

eval thm3 "?lsd_3 An,t,u,x,y,j (n>=1 & $link(x,t) & $link(y,u) &

$pref(t,u) & n<=x/9 & j+n<=y & $boolean(j,n,u)) => Ei $cmp(i,j,n,t,u)":

# returns TRUE

# 24534 ms

This returns TRUE. The expensive part of the computation is the construction of the cmp

automaton, which required 76 Gigabytes of RAM and 1886 seconds of CPU time on a Linux
machine. It has 431 states.

Remark 4. It turns out that the bound 3`−2 is optimal for n ≥ 2; this can also be proved
with Walnut. We omit the details.

The rest of the paper uses the following strategy. Instead of directly proving results about
an infinite Stewart word T (t) (which would require Büchi automata), we must often instead
restate our claims about infinite Stewart words to be about their finite prefixes generated
by a finite prefix t of t. This sometimes requires creative rephrasing involving stronger
statements. For example, instead of proving a particular kind of factor appears in T (t), we
prove that it appears provided we look at a sufficiently long prefix, where the length of the
prefix is given explicitly in terms of the size of the factor. This is where the estimate in
Theorem 3 becomes so useful. An example is Lemma 7 below.

6 Palindromic factors of the Stewart words

We are interested in what factors of Stewart words are (nonempty) palindromes. To do so,
we write a first-order logical statement for the lengths of such words.

∃t, i ∀j (j < n) =⇒ T (t)[i+ j] = T (t)[i+ n− j − 1].

When we implement this in Walnut we get

def pal "?lsd_3 Et,i Aj (j<n) => TP[t][i+j]=TP[t][(i+n)-(j+1)]":

# 4 states, 471 ms

The resulting automaton is depicted below in Figure 3. It gives the base-3 representation
of the possible lengths of palindromes occurring in the Stewart words, starting with the least
significant digit.

0

10, 1

2

2 3

0, 1, 2

0, 1

0

Figure 3: Lengths of palindromes in the Stewart words.

As you can see from inspecting this automaton, the possible lengths of palindromes are
exactly {0, 1, 2, 3, 4, 5, 6, 7}. Now we can prove

Theorem 5. The only possible palindromes occurring in the Stewart words are as follows:

{ε, 0, 1, ?, 00, 11, 010, 101, 0110, 1001, 00100, 11011, 010010, 101101, 0110110, 1001001}.

Furthermore, each such palindrome occurs in the length-81 word specified by any Stewart
pattern sequence of length 4.

7 The pattern xxyyxx

Let us consider occurrences of the pattern xxyyxx in Stewart words. Here we assume that
x is nonempty, but y is allowed to be empty.

Say that |x| = m and |y| = n. We can express the assertion that such a pattern occurs
as follows:

∃t, i,m, n (m ≥ 1) ∧
(∀j (j < m) =⇒ T (t)[i+ j] = T (t)[i+ j +m]) ∧
(∀j (j < 2m) =⇒ T (t)[i+ j] = T (t)[i+ j + 2m+ 2n] ∧
(∀j (j < n) =⇒ T (t)[i+ j + 2m] = T (t)[i+ j + 2m+ n].

The translation into Walnut is

eval xxyyxx "?lsd_3 Et,i,m,n (m>=1) &

(Aj (j<m) => TP[t][i+j]=TP[t][i+j+m]) &

(Aj (j<2*m) => TP[t][i+j]=TP[t][i+j+2*m+2*n]) &

(Aj (j<n) => TP[t][i+j+2*m]=TP[t][i+j+2*m+n])":

# returns FALSE

# 37396 ms

and Walnut returns FALSE, so there is no such pattern.

8 Critical exponent of the Stewart words

Let us recall the notions of period and exponent of a finite word. If w[i] = w[i+ p] for p ≥ 1
and 0 ≤ i < |w| − p, then w is said to have period p. The smallest period of w is denoted
per(w). The exponent of w, written exp(w), is defined to be |w|/ per(w).

Theorem 6. Let t be an infinite sequence of Stewart patterns. Then the Stewart word T (t)
has critical exponent 3, which is not attained by any finite factor.

Proof. Let us prove first that none of the infinite words contains a cube.
We write down a first-order logical formula for the property that some finite Stewart

word coded by t contains a cube, namely

∃i, p, t (p ≥ 1) ∧ ∀j (j < 2p) =⇒ T (t)[i+ j] = T (t)[i+ p+ j].

The translation into Walnut is as follows:

eval hascube "?lsd_3 Ei,p,t p>=1 & Aj (j<2*p) => TP[t][i+j]=TP[t][i+j+p]":

# returns FALSE

# 1280 ms

When we run this in Walnut, we get the answer FALSE, so there are no cubes. In fact, this
was already known, and is a consequence of a recent result of Boccuto and Carpi [3].

Next, given an infinite sequence of Stewart patterns t, we have to demonstrate the exis-
tence of factors of exponent arbitrarily close to 3. It suffices to prove the following result.

Lemma 7. For all Toeplitz words t of length ` ≥ 4 coding a length-3` word, there exists a
boolean factor of T (t) of period 3`−3 and length 3`−2 − 1, and hence of exponent 3− 33−`.

Proof. We can use Walnut to prove this, too. Again, we let x = 3|t|, so the hypothesis |t| ≥ 4
is expressed as x ≥ 81.

eval critexp "?lsd_3 At,x,p ($link(x,t) & p=x/27 & x>=81) =>

Ei i+3*p<=x & $boolean(i,3*p-1,t) & $faceq(i,i+p,2*p-1,t)":

# returns TRUE

# 116 ms

and the output from Walnut is TRUE.

This completes the proof of Theorem 6.

9 Orders of squares

As we have seen in the previous section, each infinite Stewart word contains factors of
exponent arbitrarily close to 3, and in particular squares. However, the orders of the possible
squares that can occur are greatly restricted, as the following result shows.

Theorem 8. Every infinite Stewart word contains squares xx with

|x| ∈ {3i : i ≥ 0} ∪ {2 · 3i : i ≥ 0},

and these are the only possible orders of squares that occur.

Proof. The proof has two steps. First, we compute the possible orders of squares that occur
in any Stewart word, with the following Walnut code.

def squareorder "?lsd_3 Et,i (n>=1) & $faceq(i,i+n,n,t)":

The resulting automaton is depicted in Figure 4, and clearly accepts only the base-3
representations of the form 0∗{1, 2}.

0

0

11, 2

0

Figure 4: Orders of squares that occur in Toeplitz words.

Next, we show that if we go far enough out in any given Stewart word, squares of the
given orders actually do occur. Using Theorem 3, we see that if a square of order 2 ·3i occurs,
it must occur in a prefix coded by a word of length t, where 36 · 3i ≤ 3|t|.

eval all_squares_exist "?lsd_3 An,x,t ($link(x,t) & n<=x/36 &

Ey $power3(y) & (n=y|n=2*y)) => Ei $faceq(i,i+n,n,t)":

and Walnut returns TRUE.

10 Subword complexity

In this section we discuss the subword complexity (aka factor complexity) of the infinite
Stewart words. The subword complexity function ρ(n) counts the number of distinct length-
n subwords occurring in an infinite word.

Theorem 9. Every infinite word T (t) has exactly 2n distinct factors of length n, for n ≥ 1.

Proof. A factor x of an infinite binary word w is right-special if both x0 and x1 occur in
w. To prove the theorem, we prove the equivalent result that there are exactly 2 distinct
right-special boolean factors of length n, for all n ≥ 1. Let t ∈ Aω. From Theorem 3 we
know that if a factor of length n occurs in T (t), it occurs in a prefix t of length 3dne of t. We
then prove that for each finite word t, if we look at the lengths of factors guaranteed to occur
in T (t) by Theorem 3 above, then there are always exactly two such that are right-special,
and there are never three.

First we need to define a logical formula.

• rtspec(i, n, t) expresses the assertion that T (t)[i..i+ n− 1] consists entirely of boolean
symbols and is right-special for the finite word T (t).

Next, we create assertions that there are at least two right-special factors of each length
≥ 1 and there are never three.

def rtspec "?lsd_3 $boolean(i,n,t) & Ej $boolean(j,n,t) &

(TP[t][i+n]!=TP[t][j+n]) & $faceq(i,j,n,t)":

# vars i,n,t

# 47 states, 234 ms

eval twors "?lsd_3 At,x,n ($link7(x,t) & n>=1 & n<=x/9) =>

Ei1,i2 $rtspec(i1,n,t) & $rtspec(i2,n,t) & ~$faceq(i1,i2,n,t)":

# returns TRUE

# 178 ms

eval threers "?lsd_3 Ei1,i2,i3,t,n (n>=1) & $rtspec(i1,n,t) & $rtspec(i2,n,t) &

$rtspec(i3,n,t) & ~$faceq(i1,i2,n,t) & ~$faceq(i2,i3,n,t) & ~$faceq(i1,i3,n,t)":

Here twors returns TRUE and threers returns FALSE; this completes the proof.

Thus, the Stewart words provide yet more examples of the class of sequences studied by
Rote [15].

11 Common factors

Recall that A = {a, b, c, d, e, f}. Let t,u ∈ Aω. We would like to characterize those pairs
(t,u) such that the Stewart words T (t) and T (u) contain arbitrarily large factors in common.

12



To do so we need to introduce the Hamming distance between Stewart patterns. We
define H(t, u) for t, u ∈ A as the number of positions on which T (t) and T (u) differ. For
example, H(a, b) = 2, since the patterns 01? and 10? differ in the first two positions, but
agree on the third.

Next, define

X := {(t, u) ∈ A× A : H(t, u) ∈ {0, 3}};
Y := {(t, u) ∈ A× A : H(t, u) = 2}.

Then we have the following result.

Theorem 10. The infinite Stewart words T (t) and T (u) share arbitrarily large factors in
common if and only if t × u ∈ Xω. Furthermore, if j is the smallest index on which
(t× u)[j] 6∈ X, then T (t) and T (u) have no factors in common of length > 3j+2.

Proof. For the first claim, we use the following Walnut code.

eval commonfac "?lsd_3 Ex,i1,i2 $link(x,t) & $link(x,u) & $cmp(i1,i2,x/9,t,u)":

# 4 states, 350259ms

Walnut then computes a 4-state automaton accepting those pairs of finite sequences having
common factors. In abbreviated form, this is depicted in Figure 5.

0

X

1

[0,0]

2

Y

[0,0]

[0,0]

3

X∪Y X∪Y

Figure 5: Finite sequences having common factors.

By inspection, we see that the only infinite sequences accepted by this automaton are those
in Xω.

For the second claim, we need to build an automaton that identifies the first position
where two finite words t, u of the same length differ. This can be done in Walnut as follows:

reg differ lsd_7 lsd_7 lsd_3

"([1,1,0]|[1,4,0]|[1,5,0]|[2,2,0]|[2,3,0]|[2,6,0]|

[3,2,0]|[3,3,0]|[3,6,0]|[4,1,0]|[4,4,0]|[4,5,0]|

[5,1,0]|[5,4,0]|[5,5,0]|[6,2,0]|[6,3,0]|[6,6,0])*

([1,2,1]|[1,3,1]|[1,6,1]|[2,1,1]|[2,4,1]|[2,5,1]|

[3,1,1]|[3,4,1]|[3,5,1]|[4,2,1]|[4,3,1]|[4,6,1]|

[5,2,1]|[5,3,1]|[5,6,1]|[6,1,1]|[6,4,1]|[6,5,1])

([1,1,0]|[1,2,0]|[1,3,0]|[1,4,0]|[1,5,0]|[1,6,0]|

[2,1,0]|[2,2,0]|[2,3,0]|[2,4,0]|[2,5,0]|[2,6,0]|

[3,1,0]|[3,2,0]|[3,3,0]|[3,4,0]|[3,5,0]|[3,6,0]|

[4,1,0]|[4,2,0]|[4,3,0]|[4,4,0]|[4,5,0]|[4,6,0]|

[5,1,0]|[5,2,0]|[5,3,0]|[5,4,0]|[5,5,0]|[5,6,0]|

[6,1,0]|[6,2,0]|[6,3,0]|[6,4,0]|[6,5,0]|[6,6,0])*

[0,0,0]*":

Here differ(t, u, x) is true if x = 3j and (t[j], u[j]) 6∈ X. Then we use the Walnut code

eval compare "?lsd_3 At,u,x ($link(243*x,t) & $link(243*x,u) & $differ(t,u,x))

=> Ai Aj ~$cmp(i,j,9*x+1,t,u)":

# TRUE, 14401 ms

which returns TRUE.

12 Automatic Stewart words

A paperfolding word is automatic if and only if its associated sequence of patterns is ulti-
mately periodic (see [1]). An analogous result holds for Stewart words:

Theorem 11. A Stewart word T (t) is 3-automatic if and only if t is ultimately periodic.

Proof. Suppose t is ultimately periodic. Then the language L of prefixes of t is regular. We
can then use the usual direct product construction for automata to create an automaton M ′

from M by forcing the first components of inputs to lie in L. Then M ′ demonstrates that
T (t) is 3-automatic.

On the other hand, suppose T (t) is 3-automatic and computed by a DFAO

N = (Q,Σ3,∆, δ, q0, τ).

It is easy to check that t0 := t[0] is uniquely specified by the values of T (t)[0..8]; that is, by
τ(δ(q0, x)) for |x| = 2. Next, locate the position p0 of the ? symbol in t0, and determine
q1 = δ(q0, p0). By examining the outputs of N on words of length 2, starting from the state
q1, we can then uniquely determine t1. Continuing in this fashion, we generate t element by
element. Since N has finitely many states, eventually we return to a state previously visited,
at which point t becomes periodic.

Example 12. As an example, consider the 3-automatic sequence computed by the DFAO
(lsd-first) in Figure 6.

q0/0

q1/00

q2/11

q3/0
2

0,1,2

0,1,2
0

1

2

Figure 6: A 3-automatic sequence.

Our labeling procedure labels q0 with a and then q3 with d, and then returns to state q0.
So t = (ad)ω.

13 Arithmetic progressions

We are concerned with occurrences of the words 01010 and 10101 in arithmetic progressions
in an infinite word x; this means there exist indices i,m with m ≥ 1 such that x[i] =
x[i+ 2m] = x[i+ 4m] = a and x[i+m] = x[i+ 3m] = a, for some a ∈ {0, 1}.

Theorem 13. No infinite Stewart word contains either 01010 or 10101 in arithmetic pro-
gression.

Proof. We can prove this with the following Walnut statement asserting the existence of such
an arithmetic progression.

eval arithprog "?lsd_3 Et,i,m (m>=1) & TP[t][i]=TP[t][i+2*m] &

TP[t][i]=TP[t][i+4*m] & TP[t][i+m]=TP[t][i+3*m] & TP[t][i]!=TP[t][i+m]":

# 607 ms

and Walnut returns FALSE.

For more about arithmetic progressions in Toeplitz words, see [9].

14 Going further

Everything we have done in this paper applies (with minor changes) to any finite set of
Toeplitz patterns of the same length, each containing exactly one occurrence of the symbol
?.

[5] J. Cassaigne and J. Karhumäki. Toeplitz words, generalized periodicity and periodically
iterated morphisms. European J. Combin. 18 (1997), 497–510.

[6] F. M. Dekking, M. Mendès France, and A. J. van der Poorten. Folds! Math. Intelligencer
4 (1982), 130–138, 173–181, 190–195. Erratum, 5 (1983), 5.

[7] G. Fici. Some remarks on automatic sequences, Toeplitz words and perfect
shuffling. Talk for the One World Combinatorics on Words Seminar. Avail-
able at http://www.i2m.univ-amu.fr/wiki/Combinatorics-on-Words-seminar/

_media/seminar2021:20211206fici.pdf, December 6 2021.

Appendix

This is the file TP.txt defining the Stewart automaton. The interested reader is recom-
mended to put it in the directory Word Automata Library of Walnut before typing any
commands.

lsd_7 lsd_3

0 2

0 0 -> 3

1 0 -> 1

1 1 -> 2

1 2 -> 0

2 0 -> 2

2 1 -> 1

2 2 -> 0

3 0 -> 1

3 1 -> 0

3 2 -> 2

4 0 -> 2

4 1 -> 0

4 2 -> 1

5 0 -> 0

5 1 -> 1

5 2 -> 2

6 0 -> 0

6 1 -> 2

6 2 -> 1

1 0

0 0 -> 4

0 1 -> 1

0 2 -> 1

1 0 -> 1

1 1 -> 1

1 2 -> 1

2 0 -> 1

2 1 -> 1

2 2 -> 1

3 0 -> 1

3 1 -> 1

3 2 -> 1

4 0 -> 1

4 1 -> 1

4 2 -> 1

5 0 -> 1

5 1 -> 1

5 2 -> 1

6 0 -> 1

6 1 -> 1

6 2 -> 1

2 1

0 0 -> 5

1 0 -> 2

1 1 -> 2

1 2 -> 2

2 0 -> 2

2 1 -> 2

2 2 -> 2

3 0 -> 2

3 1 -> 2

3 2 -> 2

4 0 -> 2

4 1 -> 2

4 2 -> 2

5 0 -> 2

5 1 -> 2

5 2 -> 2

6 0 -> 2

6 1 -> 2

6 2 -> 2

3 2

0 0 -> 3

4 0

0 0 -> 4

5 1

0 0 -> 5

